

(%i1)

```
/* define special summation function */
f(i,j) := sum(R[i,j,sigma,0]*gContr[i,sigma]*gContr[j,0],sigma,0,3)
        + sum(R[i,j,sigma,1]*gContr[i,sigma]*gContr[j,1],sigma,0,3)
        + sum(R[i,j,sigma,2]*gContr[i,sigma]*gContr[j,2],sigma,0,3)
        + sum(R[i,j,sigma,3]*gContr[i,sigma]*gContr[j,3],sigma,0,3);
```

(%o1) $f(i, j) := \text{sum}(R_{i, j, \sigma, 0} g_{\text{Contr } i, \sigma} g_{\text{Contr } j, 0, \sigma, 0, 3}) +$

$\text{sum}(R_{i, j, \sigma, 1} g_{\text{Contr } i, \sigma} g_{\text{Contr } j, 1, \sigma, 0, 3}) +$

$\text{sum}(R_{i, j, \sigma, 2} g_{\text{Contr } i, \sigma} g_{\text{Contr } j, 2, \sigma, 0, 3}) +$

$\text{sum}(R_{i, j, \sigma, 3} g_{\text{Contr } i, \sigma} g_{\text{Contr } j, 3, \sigma, 0, 3})$

(%i2) /* define coordinate vector */

```
array(x, 3);
[x[0],x[1],x[2],x[3]]: [t, r, theta, phi];
```

(%o2) x

(%o3) [t , r , θ , ϕ]

(%i4) /* g1 is symm. metric with indices 1...4 */

```
g1: matrix(
  [-(1-r^2/alpha^2),0,0,0],
  [0,(1-r^2/alpha^2)^-1,0,0],
  [0,0,r^2,0],
  [0,0,0,r^2*sin(theta)^2]
);
```

(%o4)
$$\begin{bmatrix} \frac{r^2}{\alpha^2} - 1 & 0 & 0 & 0 \\ 0 & \frac{1}{1 - \frac{r^2}{\alpha^2}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2 \sin(\theta)^2 \end{bmatrix}$$

(%i5) /* contravariant g is inverse of g */

```
gContr1: ratsimp(invert(g1));
```

(%o5)

$$\begin{bmatrix} \frac{\alpha^2}{r^2 - \alpha^2} & 0 & 0 & 0 \\ 0 & -\frac{r^2 - \alpha^2}{\alpha^2} & 0 & 0 \\ 0 & 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & 0 & \frac{1}{r^2 \sin(\theta)^2} \end{bmatrix}$$

(%i6)

```
/* g1 and gContr1 are transformed to g and gContr (indices 0...3) */
for mu:0 thru 3 do {
for nu:0 thru 3 do {
    g      [mu,nu]: g1      [mu+1, nu+1],
    gContr[mu,nu]: gContr1[mu+1, nu+1]
}}$
```

(%i7) /* computation of Christoffel symbols $\Gamma^{\sigma}_{\mu\nu}$ */

```
for sigma:0 thru 3 do {
for mu:0 thru 3 do {
for nu:0 thru 3 do {
    Gamma[sigma,mu,nu] :
    /* rho sum by function call: */
    sum(
        1/2 * gContr[sigma,rho]*(
            diff(g[nu,rho],x[mu]) +
            diff(g[rho,mu],x[nu]) -
            diff(g[mu,nu],x[rho])),
        rho, 0, 3),
    /* evaluate differentiation dy/dr */
    Gamma[sigma,mu,nu]: ev(Gamma[sigma,mu,nu],diff)
}}}$
```

(%i8) /* display Gamma's being different from zero */

```
for i:0 thru 3 do {
for j:0 thru 3 do {
for k:0 thru 3 do {
    if Gamma[i,j,k] # 0 then {
        display(Gamma[i,j,k])
    }}}}$
```

$$\Gamma_{0,0,1} = \frac{r}{r^2 - \alpha^2}$$

$$\Gamma_{0,1,0} = \frac{r}{r^2 - \alpha^2}$$

$$\Gamma_{1,0,0} = \frac{r(r^2 - \alpha^2)}{\alpha^4}$$

$$\Gamma_{1,1,1} = -\frac{r(r^2 - \alpha^2)}{\alpha^4 \left(1 - \frac{r^2}{\alpha^2}\right)^2}$$

$$\Gamma_{1,2,2} = \frac{r(r^2 - \alpha^2)}{\alpha^2}$$

$$\Gamma_{1,3,3} = \frac{r(r^2 - \alpha^2) \sin(\theta)^2}{\alpha^2}$$

$$\Gamma_{2,1,2} = \frac{1}{r}$$

$$\Gamma_{2,2,1} = \frac{1}{r}$$

$$\Gamma_{2,3,3} = -\cos(\theta) \sin(\theta)$$

$$\Gamma_{3,1,3} = \frac{1}{r}$$

$$\Gamma_{3,2,3} = \frac{\cos(\theta)}{\sin(\theta)}$$

$$\Gamma_{3,3,1} = \frac{1}{r}$$

$$\Gamma_{3,3,2} = \frac{\cos(\theta)}{\sin(\theta)}$$

```
(%i9) /* compute Riemann tensor elements */
for rho:0 thru 3 do {
for sigma:0 thru 3 do {
for mu:0 thru 3 do {
for nu:0 thru 3 do {
  R[rho,sigma,mu,nu] :
  diff(Gamma[rho,nu,sigma],x[mu]) -
  diff(Gamma[rho,mu,sigma],x[nu]) +
  /* lambda sums by function call: */
  sum(
    Gamma[rho,mu,lambda] * Gamma[lambda,nu,sigma] -
    Gamma[rho,nu,lambda] * Gamma[lambda,mu,sigma],
    lambda, 0, 3)
}}}}$
```

```
(%i10) /* display R's being different from zero */
for i:0 thru 3 do {
for j:0 thru 3 do {
for k:0 thru 3 do {
for l:0 thru 3 do {
  R[i,j,k,l] : /*ratsimp*/(factor(R[i,j,k,l])),
  if R[i,j,k,l] # 0 then display(R[i,j,k,l])
}}}}$
```

$$R_{0,1,0,1} = -\frac{1}{(r - \alpha)(r + \alpha)}$$

$$R_{0,1,1,0} = \frac{1}{(r-\alpha)(r+\alpha)}$$

$$R_{0,2,0,2} = \frac{r^2}{\alpha^2}$$

$$R_{0,2,2,0} = -\frac{r^2}{\alpha^2}$$

$$R_{0,3,0,3} = \frac{r^2 \sin(\theta)^2}{\alpha^2}$$

$$R_{0,3,3,0} = -\frac{r^2 \sin(\theta)^2}{\alpha^2}$$

$$R_{1,0,0,1} = -\frac{(r-\alpha)(r+\alpha)}{\alpha^4}$$

$$R_{1,0,1,0} = \frac{(r-\alpha)(r+\alpha)}{\alpha^4}$$

$$R_{1,2,1,2} = \frac{r^2}{\alpha^2}$$

$$R_{1,2,2,1} = -\frac{r^2}{\alpha^2}$$

$$R_{1,3,1,3} = \frac{r^2 \sin(\theta)^2}{\alpha^2}$$

$$R_{1,3,3,1} = -\frac{r^2 \sin(\theta)^2}{\alpha^2}$$

$$R_{2,0,0,2} = -\frac{(r-\alpha)(r+\alpha)}{\alpha^4}$$

$$R_{2,0,2,0} = \frac{(r-\alpha)(r+\alpha)}{\alpha^4}$$

$$R_{2,1,1,2} = \frac{1}{(r-\alpha)(r+\alpha)}$$

$$R_{2,1,2,1} = -\frac{1}{(r-\alpha)(r+\alpha)}$$

$$R_{2,3,2,3} = \frac{r^2 \sin(\theta)^2}{\alpha^2}$$

$$R_{2,3,3,2} = -\frac{r^2 \sin(\theta)^2}{\alpha^2}$$

$$R_{3,0,0,3} = -\frac{(r-\alpha)(r+\alpha)}{\alpha^4}$$

$$R_{3,0,3,0} = \frac{(r-\alpha)(r+\alpha)}{\alpha^4}$$

$$R_{3,1,1,3} = \frac{1}{(r-\alpha)(r+\alpha)}$$

$$R_{3,1,3,1} = -\frac{1}{(r-\alpha)(r+\alpha)}$$

$$R_{3,2,2,3} = -\frac{r^2}{\alpha^2}$$

$$R_{3,2,3,2} = \frac{r^2}{\alpha^2}$$

```
(%i11) /* Ricci tensor Ric[mu,nu] */
for mu:0 thru 3 do {
for nu:0 thru 3 do {
Ric[mu,nu]: sum(R[lambda,mu,lambda,nu], lambda, 0, 3)
}}$
```

```
(%i12) /* display Ric's being different from zero */
for i:0 thru 3 do {
for j:0 thru 3 do {
Ric[i,j] : /*ratsimp*/(factor(Ric[i,j])),
if Ric[i,j] # 0 then display(Ric[i,j])
}}$
```

$$Ric_{0,0} = \frac{3(r-\alpha)(r+\alpha)}{\alpha^4}$$

$$Ric_{1,1} = -\frac{3}{(r-\alpha)(r+\alpha)}$$

$$Ric_{2,2} = \frac{3r^2}{\alpha^2}$$

$$Ric_{3,3} = \frac{3r^2 \sin(\theta)^2}{\alpha^2}$$

```
(%i13) /* Ricci Scalar */
RicSc: sum(gContr[0,lambda]*Ric[lambda,0], lambda, 0, 3)
+ sum(gContr[1,lambda]*Ric[lambda,1], lambda, 0, 3)
+ sum(gContr[2,lambda]*Ric[lambda,2], lambda, 0, 3)
+ sum(gContr[3,lambda]*Ric[lambda,3], lambda, 0, 3)
;
```

```
(%o13) 
$$\frac{3(r^2 - \alpha^2)}{\alpha^2(r-\alpha)(r+\alpha)} + \frac{3(r-\alpha)(r+\alpha)}{\alpha^2(r^2 - \alpha^2)} + \frac{6}{\alpha^2}$$

```

```
(%i14) ratsimp(RicSc);
```

```
(%o14) 
$$\frac{12}{\alpha^2}$$

```

```
(%i15)
```

```

/* Test for R^q */
for mu: 0 thru 3 do (
for sigma:0 thru 3 do (
for nu: 0 thru 3 do (
for rho: 0 thru 3 do (
  R_q: R[mu,sigma,nu,rho] + R[mu,rho,sigma,nu] + R[mu,nu,rho,sigma],
  if R_q # 0 then (
    display("=====Einstein equation R^q=0 not fulfilled! "),
    display(mu,sigma,nu,rho),
    display(R_q)
  )
)))));

```

(%o15) done

```

(%i16) /* Raising of indices,
        contravarinat metric el. is g^x^x(contr.) = 1/g_x_x(cov.) */
/*print("Riemann elements R^0_1^0^1, R^0_2^0^2, R^0_3^0^3:");*/

```

```

R0101: f(0,1);
R0202: f(0,2);
R0303: f(0,3);

```

(%o16)
$$\frac{1}{(r - \alpha)(r + \alpha)}$$

(%o17)
$$\frac{1}{r^2 - \alpha^2}$$

(%o18)
$$\frac{1}{r^2 - \alpha^2}$$

```

(%i19) R0101: factor(R0101);
R0202: factor(R0202);
R0303: factor(R0303);

```

(%o19)
$$\frac{1}{(r - \alpha)(r + \alpha)}$$

(%o20)
$$\frac{1}{(r - \alpha)(r + \alpha)}$$

(%o21)
$$\frac{1}{(r - \alpha)(r + \alpha)}$$

```

(%i22) R1010: f(1,0);
R1212: f(1,2);
R1313: f(1,3);

```

(%o22)
$$-\frac{(r - \alpha)(r + \alpha)}{\alpha^4}$$

(%o23)
$$-\frac{r^2 - \alpha^2}{\alpha^4}$$

$$(\%024) \quad - \frac{r^2 - \alpha^2}{\alpha^4}$$

```
(%i25) R1010: factor(R1010);
      R1212: factor(R1212);
      R1313: factor(R1313);
```

$$(\%025) \quad - \frac{(r - \alpha)(r + \alpha)}{\alpha^4}$$

$$(\%026) \quad - \frac{(r - \alpha)(r + \alpha)}{\alpha^4}$$

$$(\%027) \quad - \frac{(r - \alpha)(r + \alpha)}{\alpha^4}$$

```
(%i28) R2020: f(2,0);
      R2121: f(2,1);
      R2323: f(2,3);
```

$$(\%028) \quad \frac{(r - \alpha)(r + \alpha)}{\alpha^2 r^2 (r^2 - \alpha^2)}$$

$$(\%029) \quad \frac{r^2 - \alpha^2}{\alpha^2 r^2 (r - \alpha)(r + \alpha)}$$

$$(\%030) \quad \frac{1}{\alpha^2 r^2}$$

```
(%i31) R2020: factor(R2020);
      R2121: factor(R2121);
      R2323: factor(R2323);
```

$$(\%031) \quad \frac{1}{\alpha^2 r^2}$$

$$(\%032) \quad \frac{1}{\alpha^2 r^2}$$

$$(\%033) \quad \frac{1}{\alpha^2 r^2}$$

```
(%i34) R3030: f(3,0);
      R3131: f(3,1);
      R3232: f(3,2);
```

$$(\%034) \quad \frac{(r - \alpha)(r + \alpha)}{\alpha^2 r^2 (r^2 - \alpha^2) \sin(\theta)^2}$$

$$(\%035) \quad \frac{r^2 - \alpha^2}{\alpha^2 r^2 (r - \alpha)(r + \alpha) \sin(\theta)^2}$$

$$(\%036) \quad \frac{1}{\alpha^2 r^2 \sin(\theta)^2}$$

```
(%i37) R3030: factor(R3030);
      R3131: factor(R3131);
      R3232: factor(R3232);
```

```
(%o37) 
$$\frac{1}{\alpha^2 r^2 \sin(\theta)^2}$$

```

```
(%o38) 
$$\frac{1}{\alpha^2 r^2 \sin(\theta)^2}$$

```

```
(%o39) 
$$\frac{1}{\alpha^2 r^2 \sin(\theta)^2}$$

```

```
(%i40) /* Coulomb law */
      DivE : R0101 + R0202 + R0303;
```

```
(%o40) 
$$\frac{3}{(r - \alpha)(r + \alpha)}$$

```

```
(%i41) ratsimp(DivE);
```

```
(%o41) 
$$\frac{3}{r^2 - \alpha^2}$$

```

```
(%i42) /* J[r] */
      Jr : -(R1010 + R1212 + R1313);
```

```
(%o42) 
$$\frac{3(r - \alpha)(r + \alpha)}{\alpha^4}$$

```

```
(%i43) ratsimp(Jr);
```

```
(%o43) 
$$\frac{3r^2 - 3\alpha^2}{\alpha^4}$$

```

```
(%i44) /* J[theta] */
      Jtheta : -(R2020 + R2121 + R2323);
```

```
(%o44) 
$$-\frac{3}{\alpha^2 r^2}$$

```

```
(%i45) ratsimp(Jtheta);
```

```
(%o45) 
$$-\frac{3}{\alpha^2 r^2}$$

```

```
(%i46) /* J[phi] */
      Jphi : -(R3030 + R3131 + R3232);
```

```
(%o46) 
$$-\frac{3}{\alpha^2 r^2 \sin(\theta)^2}$$

```



```
(%i47) ratsimp(Jphi);
```

```
(%o47) 
$$-\frac{3}{\alpha^2 r^2 \sin(\theta)^2}$$

```

```
(%i48) DivE_p: ev(at(DivE,[alpha=1]));
```

```
(%o48) 
$$\frac{3}{(r-1)(r+1)}$$

```

```
(%i49) Jr_p: ev(at(Jr,[alpha=1]));
```

```
(%o49) 
$$3(r-1)(r+1)$$

```

```
(%i50) Jtheta_p: ev(at(Jtheta,[alpha=1]));
```

```
(%o50) 
$$-\frac{3}{r^2}$$

```

```
(%i52) Jphi_p: ev(at(Jphi,[alpha=1, theta=%pi/2]));
```

```
(%o52) 
$$-\frac{3}{r^2}$$


```

```
(%i54)
```

```
wxplot2d([DivE_p], [r,0,3],[y,-30,30], [gnuplot_preamble, "set zeroaxis;"],
[xlabel, "r"], [ylabel, "Charge Density"])$
```

Output file "C:/Documents and Settings/Administrator/maxout.png".

```
(%t54)
```




```
(%i57)
```

```
wxplot2d([Jr_p], [r,0,3],[y,-3,20], [gnuplot_preamble, "set zeroaxis;"],
[xlabel, "r"], [ylabel, "J_r"])$
```

Output file "C:/Documents and Settings/Administrator/maxout.png".

(%t57)




(%i59)

```
wxplot2d([Jtheta_p], [r,0,3],[y,-20,0], [gnuplot_preamble, "set zeroaxis;"],
[xlabel, "r"], [ylabel, "J_theta = J_phi(theta=pi/2)"])
```

Output file "C:/Documents and Settings/Administrator/maxout.png".

(%t59)



(%i60)