

```
(%i1)
/* define special summation function */
f(i,j) := sum(R[i,j,sigma,0]*gContr[i,sigma]*gContr[j,0],sigma,0,3)
        + sum(R[i,j,sigma,1]*gContr[i,sigma]*gContr[j,1],sigma,0,3)
        + sum(R[i,j,sigma,2]*gContr[i,sigma]*gContr[j,2],sigma,0,3)
        + sum(R[i,j,sigma,3]*gContr[i,sigma]*gContr[j,3],sigma,0,3);
```

```
(%o1)  f(i , j) := sum(Ri , j , σ , 0 gContri , σ gContrj , 0 , σ , 0 , 3) +
sum(Ri , j , σ , 1 gContri , σ gContrj , 1 , σ , 0 , 3) +
sum(Ri , j , σ , 2 gContri , σ gContrj , 2 , σ , 0 , 3) +
sum(Ri , j , σ , 3 gContri , σ gContrj , 3 , σ , 0 , 3)
```

```
(%i2) /* define coordinate vector */
array(x, 3);
[x[0],x[1],x[2],x[3]]: [t, r, theta, phi];
```

```
(%o2) x
```

```
(%o3) [ t , r , θ , φ ]
```

```
(%i4) /* define coordinate dependent functions */
depends([a], [t]);
```

```
(%o4) [ a(t) ]
```

```
(%i5) /* g1 is symm. metric with indices 1...4 */
g1: matrix(
  [-1,0,0,0],
  [0,a^2/(1-k*r^2),0,0],
  [0,0,a^2*r^2,0],
  [0,0,0,a^2*r^2*sin(theta)^2]
);
```

```
(%o5) 
$$\begin{bmatrix} -1 & 0 & 0 & 0 \\ 0 & \frac{a^2}{1 - k r^2} & 0 & 0 \\ 0 & 0 & a^2 r^2 & 0 \\ 0 & 0 & 0 & a^2 r^2 \sin(\theta)^2 \end{bmatrix}$$

```

```
(%i6) /* contravariant g is inverse of g */
gContr1: ratsimp(invert(g1));
```

$$(\%o6) \begin{bmatrix} -1 & 0 & 0 & 0 \\ 0 & -\frac{k r^2 - 1}{a^2} & 0 & 0 \\ 0 & 0 & \frac{1}{a^2 r^2} & 0 \\ 0 & 0 & 0 & \frac{1}{a^2 r^2 \sin(\theta)^2} \end{bmatrix}$$

(%i7)

```
/* g1 and gContr1 are transformed to g and gContr (indices 0...3) */
for mu:0 thru 3 do {
for nu:0 thru 3 do {
    g [mu,nu]: g1 [mu+1, nu+1],
    gContr[mu,nu]: gContr1[mu+1, nu+1]
}}$
```

```
(%i8) /* computation of Christoffel symbols Gamma^sigma_mu_nu */
for sigma:0 thru 3 do {
for mu:0 thru 3 do {
for nu:0 thru 3 do {
    Gamma[sigma,mu,nu] :
    /* rho sum by function call: */
    sum(
        1/2 * gContr[sigma,rho]*(
            diff(g[nu,rho],x[mu] ) +
            diff(g[rho,mu],x[nu] ) -
            diff(g[mu,nu] ,x[rho])),
        rho, 0, 3),
    /* evaluate differentiation dy/dr */
    Gamma[sigma,mu,nu]: ev(Gamma[sigma,mu,nu],diff)
}}}$
```

```
(%i9) /* display Gamma's being different from zero */
for i:0 thru 3 do {
for j:0 thru 3 do {
for k:0 thru 3 do {
    if Gamma[i,j,k] # 0 then {
        display(Gamma[i,j,k])
    }}}}$
```

$$\Gamma_{0,1,1} = \frac{a \left(\frac{d}{dt} a \right)}{1 - k r^2}$$

$$\Gamma_{0,2,2} = a \left(\frac{d}{dt} a \right) r^2$$

$$\Gamma_{0,3,3} = a \left(\frac{d}{dt} a \right) r^2 \sin(\theta)^2$$

$$\Gamma_{1,0,1} = - \frac{\left(\frac{d}{dt}a\right)(kr^2 - 1)}{a(1 - kr^2)}$$

$$\Gamma_{1,1,0} = - \frac{\left(\frac{d}{dt}a\right)(kr^2 - 1)}{a(1 - kr^2)}$$

$$\Gamma_{1,1,1} = - \frac{kr(kr^2 - 1)}{(1 - kr^2)^2}$$

$$\Gamma_{1,2,2} = r(kr^2 - 1)$$

$$\Gamma_{1,3,3} = r(kr^2 - 1) \sin(\theta)^2$$

$$\Gamma_{2,0,2} = \frac{\frac{d}{dt}a}{a}$$

$$\Gamma_{2,1,2} = \frac{1}{r}$$

$$\Gamma_{2,2,0} = \frac{\frac{d}{dt}a}{a}$$

$$\Gamma_{2,2,1} = \frac{1}{r}$$

$$\Gamma_{2,3,3} = - \cos(\theta) \sin(\theta)$$

$$\Gamma_{3,0,3} = \frac{\frac{d}{dt}a}{a}$$

$$\Gamma_{3,1,3} = \frac{1}{r}$$

$$\Gamma_{3,2,3} = \frac{\cos(\theta)}{\sin(\theta)}$$

$$\Gamma_{3,3,0} = \frac{\frac{d}{dt}a}{a}$$

$$\Gamma_{3,3,1} = \frac{1}{r}$$

$$\Gamma_{3,3,2} = \frac{\cos(\theta)}{\sin(\theta)}$$

```
(%i10) /* compute Riemann tensor elements */
for rho:0 thru 3 do {
for sigma:0 thru 3 do {
for mu:0 thru 3 do {
for nu:0 thru 3 do {
  R[rho,sigma,mu,nu] :
  diff(Gamma[rho,nu,sigma],x[mu]) -
  diff(Gamma[rho,mu,sigma],x[nu]) +
  /* lambda sums by function call: */
  sum(
    Gamma[rho,mu,lambda] * Gamma[lambda,nu,sigma] -
    Gamma[rho,nu,lambda] * Gamma[lambda,mu,sigma],
    lambda, 0, 3)
}}}}$
```

```
(%i11) /* display R's being different from zero */
for i:0 thru 3 do {
for j:0 thru 3 do {
for k:0 thru 3 do {
for l:0 thru 3 do {
  R[i,j,k,l] : /*ratsimp*/(factor(R[i,j,k,l])),
  if R[i,j,k,l] # 0 then display(R[i,j,k,l])
}}}}$
```

$$R_{0,1,0,1} = - \frac{a \left(\frac{d^2}{dt^2} a \right)}{k r^2 - 1}$$

$$R_{0,1,1,0} = \frac{a \left(\frac{d^2}{dt^2} a \right)}{k r^2 - 1}$$

$$R_{0,2,0,2} = a \left(\frac{d^2}{dt^2} a \right) r^2$$

$$R_{0,2,2,0} = - a \left(\frac{d^2}{dt^2} a \right) r^2$$

$$R_{0,3,0,3} = a \left(\frac{d^2}{dt^2} a \right) r^2 \sin(\theta)^2$$

$$R_{0,3,3,0} = - a \left(\frac{d^2}{dt^2} a \right) r^2 \sin(\theta)^2$$

$$R_{1,0,0,1} = \frac{\frac{d^2}{dt^2} a}{a}$$

$$R_{1,0,1,0} = - \frac{\frac{d^2}{dt^2} a}{a}$$

$$R_{1, 2, 1, 2} = \left(k + \left(\frac{d}{dt} a \right)^2 \right) r^2$$

$$R_{1, 2, 2, 1} = - \left(k + \left(\frac{d}{dt} a \right)^2 \right) r^2$$

$$R_{1, 3, 1, 3} = \left(k + \left(\frac{d}{dt} a \right)^2 \right) r^2 \sin(\theta)^2$$

$$R_{1, 3, 3, 1} = - \left(k + \left(\frac{d}{dt} a \right)^2 \right) r^2 \sin(\theta)^2$$

$$R_{2, 0, 0, 2} = \frac{\frac{d^2}{dt^2} a}{a}$$

$$R_{2, 0, 2, 0} = - \frac{\frac{d^2}{dt^2} a}{a}$$

$$R_{2, 1, 1, 2} = \frac{k + \left(\frac{d}{dt} a \right)^2}{k r^2 - 1}$$

$$R_{2, 1, 2, 1} = - \frac{k + \left(\frac{d}{dt} a \right)^2}{k r^2 - 1}$$

$$R_{2, 3, 2, 3} = \left(k + \left(\frac{d}{dt} a \right)^2 \right) r^2 \sin(\theta)^2$$

$$R_{2, 3, 3, 2} = - \left(k + \left(\frac{d}{dt} a \right)^2 \right) r^2 \sin(\theta)^2$$

$$R_{3, 0, 0, 3} = \frac{\frac{d^2}{dt^2} a}{a}$$

$$R_{3, 0, 3, 0} = - \frac{\frac{d^2}{dt^2} a}{a}$$

$$R_{3, 1, 1, 3} = \frac{k + \left(\frac{d}{dt} a \right)^2}{k r^2 - 1}$$

$$R_{3, 1, 3, 1} = - \frac{k + \left(\frac{d}{dt} a \right)^2}{k r^2 - 1}$$

$$R_{3, 2, 2, 3} = - \left(k + \left(\frac{d}{dt} a \right)^2 \right) r^2$$

$$R_{3,2,3,2} = \left(k + \left(\frac{d}{dt} a \right)^2 \right) r^2$$

```
(%i12) /* Ricci tensor Ric[mu,nu] */
for mu:0 thru 3 do {
for nu:0 thru 3 do {
Ric[mu,nu]: sum(R[lambda,mu,lambda,nu], lambda, 0, 3)
}}$
```

```
(%i13) /* display Ric's being different from zero */
for i:0 thru 3 do {
for j:0 thru 3 do {
Ric[i,j] : /*ratsimp*/(factor(Ric[i,j])),
if Ric[i,j] # 0 then display(Ric[i,j])
}}$
```

$$Ric_{0,0} = - \frac{3 \left(\frac{d^2}{dt^2} a \right)}{a}$$

$$Ric_{1,1} = - \frac{2k + a \left(\frac{d^2}{dt^2} a \right) + 2 \left(\frac{d}{dt} a \right)^2}{k r^2 - 1}$$

$$Ric_{2,2} = \left(2k + a \left(\frac{d^2}{dt^2} a \right) + 2 \left(\frac{d}{dt} a \right)^2 \right) r^2$$

$$Ric_{3,3} = \left(2k + a \left(\frac{d^2}{dt^2} a \right) + 2 \left(\frac{d}{dt} a \right)^2 \right) r^2 \sin(\theta)^2$$

```
(%i14) /* Ricci Scalar */
RicSc: sum(gContr[0,lambda]*Ric[lambda,0], lambda, 0, 3)
+ sum(gContr[1,lambda]*Ric[lambda,1], lambda, 0, 3)
+ sum(gContr[2,lambda]*Ric[lambda,2], lambda, 0, 3)
+ sum(gContr[3,lambda]*Ric[lambda,3], lambda, 0, 3)
;
```

```
(%o14) 
$$\frac{3 \left( 2k + a \left( \frac{d^2}{dt^2} a \right) + 2 \left( \frac{d}{dt} a \right)^2 \right)}{a^2} + \frac{3 \left( \frac{d^2}{dt^2} a \right)}{a}$$

```

```
(%i15) ratsimp(RicSc);
```

```
(%o15) 
$$\frac{6k + 6a \left( \frac{d^2}{dt^2} a \right) + 6 \left( \frac{d}{dt} a \right)^2}{a^2}$$

```

```
(%i16)
```

```

/* Test for R^q */
for mu: 0 thru 3 do (
for sigma:0 thru 3 do (
for nu: 0 thru 3 do (
for rho: 0 thru 3 do (
  R_q: R[mu,sigma,nu,rho] + R[mu,rho,sigma,nu] + R[mu,nu,rho,sigma],
  if R_q # 0 then (
    display("====Einstein equation R^q=0 not fulfilled! "),
    display(mu,sigma,nu,rho),
    display(R_q)
  )
)))));

```

(%o16) done

```

(%i17) /* Raising of indices,
        contravariant metric el. is g^x^x(contr.) = 1/g_x_x(cov.) */
/*print("Riemann elements R^0_1^0^1, R^0_2^0^2, R^0_3^0^3:");*/

```

R0101: f(0,1);

R0202: f(0,2);

R0303: f(0,3);

(%o17)
$$-\frac{\frac{d^2}{dt^2}a}{a}$$

(%o18)
$$-\frac{\frac{d^2}{dt^2}a}{a}$$

(%o19)
$$-\frac{\frac{d^2}{dt^2}a}{a}$$

```

(%i20) R0101: factor(R0101);
        R0202: factor(R0202);
        R0303: factor(R0303);

```

(%o20)
$$-\frac{\frac{d^2}{dt^2}a}{a}$$

(%o21)
$$-\frac{\frac{d^2}{dt^2}a}{a}$$

(%o22)
$$-\frac{\frac{d^2}{dt^2}a}{a}$$

```

(%i23) R1010: f(1,0);
        R1212: f(1,2);
        R1313: f(1,3);

```

$$(\%o23) \quad - \frac{\left(\frac{d^2}{dt^2}a\right)(kr^2 - 1)}{a^3}$$

$$(\%o24) \quad - \frac{\left(k + \left(\frac{d}{dt}a\right)^2\right)(kr^2 - 1)}{a^4}$$

$$(\%o25) \quad - \frac{\left(k + \left(\frac{d}{dt}a\right)^2\right)(kr^2 - 1)}{a^4}$$

(%i26) R1010: factor(R1010);
 R1212: factor(R1212);
 R1313: factor(R1313);

$$(\%o26) \quad - \frac{\left(\frac{d^2}{dt^2}a\right)(kr^2 - 1)}{a^3}$$

$$(\%o27) \quad - \frac{\left(k + \left(\frac{d}{dt}a\right)^2\right)(kr^2 - 1)}{a^4}$$

$$(\%o28) \quad - \frac{\left(k + \left(\frac{d}{dt}a\right)^2\right)(kr^2 - 1)}{a^4}$$

(%i29) R2020: f(2,0);
 R2121: f(2,1);
 R2323: f(2,3);

$$(\%o29) \quad \frac{\frac{d^2}{dt^2}a}{a^3 r^2}$$

$$(\%o30) \quad \frac{k + \left(\frac{d}{dt}a\right)^2}{a^4 r^2}$$

$$(\%o31) \quad \frac{k + \left(\frac{d}{dt}a\right)^2}{a^4 r^2}$$

(%i32) R2020: factor(R2020);
 R2121: factor(R2121);
 R2323: factor(R2323);

$$(\%o32) \quad \frac{\frac{d^2}{dt^2}a}{a^3 r^2}$$

$$(\%033) \quad \frac{k + \left(\frac{d}{dt} a \right)^2}{a^4 r^2}$$

$$(\%034) \quad \frac{k + \left(\frac{d}{dt} a \right)^2}{a^4 r^2}$$

```
(%i35) R3030: f(3,0);
       R3131: f(3,1);
       R3232: f(3,2);
```

$$(\%035) \quad \frac{\frac{d^2}{dt^2} a}{a^3 r^2 \sin(\theta)^2}$$

$$(\%036) \quad \frac{k + \left(\frac{d}{dt} a \right)^2}{a^4 r^2 \sin(\theta)^2}$$

$$(\%037) \quad \frac{k + \left(\frac{d}{dt} a \right)^2}{a^4 r^2 \sin(\theta)^2}$$

```
(%i38) R3030: factor(R3030);
       R3131: factor(R3131);
       R3232: factor(R3232);
```

$$(\%038) \quad \frac{\frac{d^2}{dt^2} a}{a^3 r^2 \sin(\theta)^2}$$

$$(\%039) \quad \frac{k + \left(\frac{d}{dt} a \right)^2}{a^4 r^2 \sin(\theta)^2}$$

$$(\%040) \quad \frac{k + \left(\frac{d}{dt} a \right)^2}{a^4 r^2 \sin(\theta)^2}$$

```
(%i41) /* Coulomb law */
       DivE : R0101 + R0202 + R0303;
```

$$(\%041) \quad - \frac{3 \left(\frac{d^2}{dt^2} a \right)}{a}$$

```
(%i42) ratsimp(DivE);
```

$$(\%042) \quad - \frac{3 \left(\frac{d^2}{d t^2} a \right)}{a}$$

```
(%i43) /* J[r] */
Jr : -(R1010 + R1212 + R1313);
```

$$(\%043) \quad \frac{2 \left(k + \left(\frac{d}{d t} a \right)^2 \right) (k r^2 - 1)}{a^4} + \frac{\left(\frac{d^2}{d t^2} a \right) (k r^2 - 1)}{a^3}$$

```
(%i44) ratsimp(Jr);
```

$$(\%044) \quad \frac{\left(2 k^2 + \left(a \left(\frac{d^2}{d t^2} a \right) + 2 \left(\frac{d}{d t} a \right)^2 \right) k \right) r^2 - 2 k - a \left(\frac{d^2}{d t^2} a \right) - 2 \left(\frac{d}{d t} a \right)^2}{a^4}$$

```
(%i45) /* J[theta] */
Jtheta : -(R2020 + R2121 + R2323);
```

$$(\%045) \quad - \frac{2 \left(k + \left(\frac{d}{d t} a \right)^2 \right)}{a^4 r^2} - \frac{\frac{d^2}{d t^2} a}{a^3 r^2}$$

```
(%i46) ratsimp(Jtheta);
```

$$(\%046) \quad - \frac{2 k + a \left(\frac{d^2}{d t^2} a \right) + 2 \left(\frac{d}{d t} a \right)^2}{a^4 r^2}$$

```
(%i47) /* J[phi] */
Jphi : -(R3030 + R3131 + R3232);
```

$$(\%047) \quad - \frac{2 \left(k + \left(\frac{d}{d t} a \right)^2 \right)}{a^4 r^2 \sin(\theta)^2} - \frac{\frac{d^2}{d t^2} a}{a^3 r^2 \sin(\theta)^2}$$

```
(%i48) ratsimp(Jphi);
```

$$(\%048) \quad - \frac{2 k + a \left(\frac{d^2}{d t^2} a \right) + 2 \left(\frac{d}{d t} a \right)^2}{a^4 r^2 \sin(\theta)^2}$$

```
(%i49) a: 1;
```

```
(%o49) 1
```

```
(%i50) DivE_p: ev(at(DivE,[a=1, k=1]));
```

```
(%o50) 0
```

```
(%i51) Jr_p: ev(at(Jr,[a=1, k=1]));
```

```
(%o51) 2 (r2 - 1)
```

```
(%i52) Jtheta_p: ev(at(Jtheta,[a=1, k=1, theta=%pi/2]));
```


```
(%o52) -  $\frac{2}{r^2}$ 
```

```
(%i53)
```

```
wxplot2d([Jr_p], [r,.1,5],[y,-2,50], [gnuplot_preamble, "set zeroaxis;"],  
[xlabel, "r"], [ylabel, "J_r"])$
```

Output file "C:/Documents and Settings/Administrator/maxout.png".

```
(%t53)
```




```
(%i54)
```

```
wxplot2d([Jtheta_p], [r,.1,5],[y,-20,0], [gnuplot_preamble, "set zeroaxis;"],  
[xlabel, "r"], [ylabel, "J_theta = J_phi"])$
```

Output file "C:/Documents and Settings/Administrator/maxout.png".

```
(%t54)
```



(%i55)